
1. はじめに 
 慶應義塾⼤学湘南藤沢キャンパスで開設されている
「スクリプト⾔語プログラミング基礎」という科⽬の秋
学期のクラスにおいて、この 2 年間、AppleScript(1)と
Smile2(2)を利⽤した数値解析のプログラミングの仕⽅に
ついて触れてきた。AppleScript ⾃体は、HyperTalk に続
く、だいぶ古い⾔語ではあるが、mac OS 上での定型処理
の⾃動化では、未だに現役で使われている。Smile2 の開
発環境上で AppleScript と Smile2 で⽤意されているライ
ブラリと組み合わせることによって、Python で numpy, 
scipy, matplotlib を利⽤した数値解析と同等の環境を構築
することができる。また、Smile2 ⾃体は独⾃のグラフィ
ックスライブラリを持っており、それを利⽤したGUI の
プログラムも作成することが可能になっている。実際に
授業中で実践した内容に基づき、Smile2 を使ってどのよ
うなプログラミングが可能になるかを紹介する。 
 
2. Smile/Smile2について 
2.1 Smile/Smile2 のリリース 

Smile の最初の版は、AppleScript のリリースにあわせ
て、1995 年、フランスのSatImage(2)という開発会社から
出されている。Smile は、SMI Limited Edition の頭⽂字
をとったもので、SMI は、Scriptable Measurements on 
Images の頭⽂字をとったものとなっている。Smile の
AppleScript 開発環境の上に、2D, 3D グラフィックスお
よびグラフのプロットなどを含む、グラフィックによる
視覚化を⾏なったもの SmileLab という製品になってい
る。Smile は、AppleScript の⾔語のバージョンが 2.0 以
降になったことを受けて作り直しており、64bit 版の
Smile2 がベータ版としてリリースされている。SatImage
のホームページでは、まだ32bit 版のSmile という名前の
ままになっている。現在は、SatImage によるベータ版の
Smile2 のサポートは、個⼈ベースになっている。1990 年
代の Smile 開発当初は有料のパッケージであったが、現
在は Smile および Smile2 共に無料で macOS 上にフリー
にダウンロードすることができる。なお、Smile2 のバー
ジョンは、2024 年6⽉現在で、beta-4.0.0 版(232)となっ
ており、ユーザからバグなどの報告がなされると、ビルド
を新しくしている。実際に、2022 年の段階で、⾏列演算
でのコマンドと Suite の組み合わせでエラーが⽣じてい

たので、連絡したら、すぐに修正した版をリリースしてく
れた経緯がある。Smile2は最新のmacOSでも稼働する。 

 

 
図1  Smile2 の稼働画⾯ 

 
2.2 SatImageOSAX 

AppleScript は、AppleScript 2.4ぐらいまでは、OSAX
（Open Scripting Architecture eXtensions）という形で、
サードパーティーがコマンドを拡張して使うことができ
る環境が⽤意されていた。しかしながら、それ2.4 以降の
版では、OSAXとして標準の Scripting Additions 以外は
使えないようになってしまった。それまで、SatImage は、
SatImageOSAX という名称で、Smile で⽤意されている
コマンド群（Suite と呼ばれる）の⼀部を、無償で開放し
てきた。SatImageOSAXは、AppleScript に必要不可⽋だ
ったものだが、それが使⽤できなくなってしまったので、
現在は⽶国のLate Night Software 社のScript Debugger(3)
のコミュニティにおいて、移植されたアプリケーション
版（SatImageOSAX.app）が⽤意されており、AppleScript
から直接利⽤することができるが、これは配列や⾏列関
係の Suite の⼀部コマンドが使えない状態になっている。
このバグに連絡はしたが、修正して貰えていない。授業で
は、その部分はSmile2 で実習を⾏なうことにして、それ
以外のSuite のコマンドを利⽤することにした。 
 
2.3 Smile2 のSuite 群 

Smile2 の SmileLab における Suite 群は、上記の
SatImageOSAX 群の Suite と合わせて、様々なコマンド
を内包するSuite が⽤意されている。以下のリストで、⼤
⽂字・⼩⽂字の使い⽅は、Smile2 のコマンド辞書にあわ
せている。 



 
• SatImageOSAX のSuite 群 
 - SatImage Text Additions 
  - SatImage File Additions 
  - Mathematical Functions 
  - Arrays 
  - Arrays and List Utilities 
• Smile Suite 群 
  - Smile 
  - Smile text suite 
  - Smile files and folder suite 
  - Smile scripted scripting 
  - Smile drawings Suite 
  - SmileLab Suite 
  - Smile over IP 
  - Smile IO devices 
  - Web Suite 
• Graphics Library 群 
  - Prologue 
  - Context 
  - Paths 
  - Text 
  - Image 
  - Layers 
  - References 
  - Miscellaneous 
• Numerics 群 
  - Linear algebra 
  - FFT and convolution 
  - Image files 
  - 3D array handling 
• Geometry 群 
  - Geometry suite 
  - Transformations 
  - Constructions 
  - Triangles 
  - Drawings 
• XMLLib 群 
  - SatImage XML DOM 
  - XML XPath, XSLT 
  - XML Documents 
  - XML Validation 
  - SatImage Property List Additions 
  - SatImage Pool Additions 
  - XNF Additions 
 

授業においては、この中のSatImageOSAX群とNumerics
群、および Smile群の中の SmileLab Suite を⽤いて、数
値計算、数値解析関係の内容を扱ったが、今後は⼀般の
2D グラフィックスについても、扱いたいと考えている。 
 
3. Smile2を⽤いた数値計算・数値解析 
3.1 配列の計算 

AppleScript は、Python の⽤に演算⼦に対しての拡張を

記述できないため、すべてコマンドで記述しなければな
らない。Python のnumpy のarray に対する要素ごとの加
減乗除は、演算⼦ではなく、Arrays の Suite の中にある、
addlist, sublist, multlist, divlist を使って⾏なう。以下は、
array を作って、各要素に cos, sin を適⽤した後、スカラ
ー積を求める例になっている。s2 の各要素には、a の各要
素n の2sin( n )cos( n )の値が格納されている。 

 
set a to createarray 10000 range {0, 2*pi} 
set c to (cos a) 
set s to (sin a) 
set cs to multlist c with s 
set s2 to multlist cs with 2 
 
リストに対しての統計計算は、statlist コマンドによっ

て、最⼤値・最⼩値（およびそのインデックス）・総和・
平均・分散・標準偏差が⼊ったstatrecord という名前のレ
コードに求めることができる。 

 
set a to {0, 1, 2, 3, 4, 5, 6} 
log (statlist a) 
-- {maximum:6.0, maximum index:7,  

     minimum:0.0, minimum index:1, 
    sum:21.0, mean:3.0, variance:4.0, stdev:2.0} 
 
また、乱数によるリストの初期化とヒストグラムの作

成も⾏なうことができる。 
 
set a to randomarray 100 range {0, 10} 
set b to histogram a samples 5 minimum 0 maximum 10 
log (b as list of real) 
-- {{1.0, 3.0, 5.0, 7.0, 9.0}, {25.0, 20.0, 18.0, 16.0, 21.0}} 

  -- 最初のリストが各区間の中央値、次のリストが個数 
 

3.2 線形代数（⾏列）の計算 
⾏列について、要素ごとの和差積商を求めるには、リス

トのaddlist, sublist, multlist, divlist がそのまま使える。加
えて、⾏列の積を求めるmultimatrixや、転置⾏列を求め
る transpose、逆⾏列を求める invertmatrixのコマンドが
ある。 

 
set x to {3, 1, 2, 2, 1, 7, 4, 5, -1} 
set m to {class:matrix, ncols:3, nrows:3, array of real:x} 
set tm to transpose m 
set im to invertmatrix m 
set ident to multmatrix m with im -- 単位⾏列になる 
 
正⽅⾏列の⾏列式を求めるcompute determinant、固有

値（複素数のものも含む）および各固有値に対応した固有
ベクトルを求める compute eigenvalues、LU分解を⾏な



う LUdecomposition などのコマンドが⽤意されている。 
 

set mat to {class:matrix, ncols:3, nrows:3, array of real:  
     ({{7, 6, 9}, {-7, 2, 2}, {4, 3, 2}}) as array of real} 

log (compute determinant mat) 
set eugenrec to compute eigenvalues mat with eigenvects 
log eigenvalues of eugenrec 
log eigenvectors of eugenrec 
set lapack to LUdecomposition mat 
log uppermatrix of lapack 
log lowermatrix of lapack 
 
連⽴⼀次⽅程式の解を求める、すなわち、AX = B（ただ

しXは正⽅⾏列、およびBは列ベクトル）となるような
⾏ベクトルA を求める solve linear systemコマンドも⽤
意されている。 

 
set mat to {class:matrix, ncols:3, nrows:3, array of real:  

      ({{1, 3, 4}, {-5, 2, 2}, {4, 3, 2}}) as array of real} 
set vecb to {5, 2, 4} as array of real 
log (solve linear system mat RHS vecb) 

 
3.3 ⾼速フーリエ変換・補間・曲線回帰の計算 
⾼速フーリエ変換は、⾳声データなどに適⽤させるこ

とができる fft1dコマンドと、画像データなどに適⽤させ
ることができる fft2dコマンドが⽤意されている。両コマ
ンド共に逆フーリエ変換を⾏なうinverseオプションが⽤
意されている。 

 
set x to createarray 144 range {-pi, pi} 
set y to randomarray 144 range {-0.75, 1} 
set ys to runningsum y 
set {kreal, kimag} to fft1d ys 

 
 補間を線形補間およびスプライン補間のどちらかを選
択して⾏なえる（通常はスプライン補間）interpolateコマ
ンドが⽤意されている。 
 

set x to runningsum (randomarray 10) 
set y to runningsum x 
set xsample to {1, 2, 3, 4, 5, 6} as array of real 
set yvalues to interpolate {x, y} at xsample 
log yvalues as list of real 
 

 多項式による曲線回帰（curve fitting）を⾏なう
fitpolynomial コマンドが⽤意されている。多項式の次数
をdegreeオプションで指定することができる。 
 

set x to runningsum (randomarray 10) 
set y to runningsum x 

set fitrec to fitpolynomial {x, y} degree 3 number 
formatting "%.3g" 
log (fit string of fitrec) 
--"2.23-2.58*x+2.07*x^2-0.113*x^3" 乱数なので 
--実⾏ごとに毎回変わるが、3次多項式で記述 

 
 その他に畳み込み（convolution）を⾏なう filterコマン
ド（2次元）、filterarray（1次元）、convolveコマンド、お
よびcorrelateコマンドが⽤意されている。また、3D array 
handling のSuite には、3次元配列（テンソル）に関して
のコマンドも⽤意されている。 

 
3.4 グラフのプロット 

グラフのプロットは、Python の matplotlib と同様のグ
ラフ作成が、Smile drawings Suite とSmileLab Suite の中
に、クラスやコマンドが⽤意されている。x座標とy座標
をリストで指定した折れ線グラフは、以下のように記述
することができる。 

 
set w to make new graphic window 
set v to make new plot view at w 
set c to make new curve at v 
set xdata of c to {1, 2, 3, 4, 5} 
set ydata of c to {1, 4, 9, 16, 25} 
draw w 

 
この⽅法で作成するよりも簡便に以下のQuickCurve関

数を使って作成することができる。この関数は、
QuickPlotLib（内部的には Smile2 の Class Scripts の
Context additions にある）ライブラリとして収められて
おり、3D のグラフも含む様々なユーティリティ関数が収
められている。上記と同じ内容をこの関数を利⽤して、記
述すると以下の1 ⾏になる。 

 
QuickCurve({1, 2, 3, 4, 5}, {1, 4, 9, 16, 25}, 0) 
 
最後の 0（または null で指定する）は、トップの plot 
viewを作って表⽰する指定で、このパラメータで、包含
されるplot viewをすることもできる。それを使えば、以
下のように複数のグラフを1 つのplot viewの中に描画す
ることができる。 

 
set w to make new graphic window 
set v to make new plot view at w 
set a to createarray 10000 range {0, 2 * pi} 
set {c, s} to {(cos a), (sin a)} 
QuickCurve(a, c, v) 
QuickCurve(a, s, v) 
 
 



 
図2 プロットされたグラフ 

 
4. Smile2でのグラフィックス描画・他 
4.1 幾何図形の描画 

Smile2 のGraphics Library のSuite 群においては、
2 次元ベクトルの演算も含む、幾何図形の描画⽤の
クラスやコマンドが収められている。これは、簡易
描画関数のライブラリになっており、例えば、以下
のスクリプトは、塗り潰された円を表⽰する。 

 
set w to BeginFigure(0) 
set {pagewidth of w, pageheight of w} to {500, 500} 
SetFillColor({0, 1, 1, 1}) -- {R, G, B, A} 
SetPenColor({1, 0.5, 0.7, 1}) 
CirclePath({250, 250}, 50) 
DrawPath("fill-stroke") 
EndFigure() 

 
図3 描画された内容 

 
4.2 GUI コンポーネント 

Smile の中では、ボタンなどの独⾃のコンポーネントを
持つダイアログを作成し、ボタンが押されたら、付随した
スクリプト中の対応するコールバック関数が呼ばれて、
処理をする GUI(Graphic User Interface)を作成する機能
がある。しかしながら、Smile2 ではまだ未実装になって
いる。 

 
4.3 XML によるDOM の操作 

Smile2は、XMLによるDOM（Document Object Model）
のスクリプトを組むことができる。以下は、SatImage の
XML に関してのチュートリアルファイルを読み込んで、
XML の DOM を解析し、<tr>のタグの中に書かれたテ
キストをログに表⽰するものである。このスクリプトは、
SatimageOSAX.app に対しても実⾏することができる。
Smile2 においては、XMLLib の Suite群の中のクラスや

コマンドを利⽤して記述されている。 
 
set xf to XMLOpen "http://www.satimage.fr/" & 
"software/en/smile/xml/xml_tutorial.html" 

  encoding «class utf8» 
set theroot to XMLRoot xf 
set theBody to XMLChild theroot index 2 
set theTables to XMLFind theBody  

                name "table" with all occurrences 
repeat with tb in theTables 
   set tableRecords to XMLFind tb  

                 name "tr" with all occurrences 
   repeat with tr in tableRecords 
      log (XMLGetText tr) 
   end repeat 
end repeat 
XMLClose xf 

 
このように、Smile2およびSatimageOSAXでは、Python

のBeautiful Soap のようにWeb から、HTMLやXMLデ
ータを読み込み、その内容をスクリプトに取り込むこと
が可能になっている。 

 
5. おわりに 

Python2 がリリースされて、numpy, scipy, matplotlib が
出てくるまでは、個⼈的にSmileLab/Smile を利⽤してき
た経緯がある。数値計算・数値解析がPython で記述でき
るようになって、だいぶ楽になったが、AppleScript でも
同じことができるということを、これからのプログラミ
ング教育で伝えていこうと考えている。Python も元はそ
うだが、AppleScript は、余分な記号を極⼒廃したプログ
ラムを記述でき、C ⾔語系列のプログラミング⾔語の思
想と対局にあるように思える。1 ⾏から記述できるインタ
ープリタをベースに、数値計算や解析が、英語のように記
述できる魅⼒は捨てがたいものがある。Smile2 はまだベ
ータ版であり、Smile で実装されていた機能の⼀部は未実
装になっているが、今後の精⼒的な移⾏のための開発と
処理の安定化に期待したい。 

